[bookmark: _GoBack]ΠΡΟΓΡΑΜΜΑ ΠΟΔΟΣΦΑΙΡΟ
ΜΕΤΑΒΛΗΤΕΣ
  ΑΚΕΡΑΙΕΣ: Ι, Ξ, ΑΓΩΝΕΣ, ΙΣΟΠΑΛΙΕΣ, ΙΣΟΠΑΛΙΕΣΕΔ[5], ΙΣΟΠΑΛΙΕΣΕΚΤΟΣ[5], ΒΑΘΜ[5], Λ, Μ, Β, Θ
  ΧΑΡΑΚΤΗΡΕΣ: ΑΠΟΤ[5, 5], ΟΝ[5], Τ, ΟΝ1, ΟΝ2
ΑΡΧΗ
  ΓΙΑ Ι ΑΠΟ 1 ΜΕΧΡΙ 5
    ΓΙΑ Ξ ΑΠΟ 1 ΜΕΧΡΙ 5
      ΑΠΟΤ[Ι, Ξ] <- "0"
    ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
  ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
  ΑΓΩΝΕΣ <- 0
  ΙΣΟΠΑΛΙΕΣ <- 0
  ΓΙΑ Ι ΑΠΟ 1 ΜΕΧΡΙ 5
    ΓΡΑΨΕ "ΔΩΣΕ ΜΟΥ ΟΝΟΜΑ ", Ι, "ης ΟΜΑΔΑΣ"
    ΔΙΑΒΑΣΕ ΟΝ[Ι] 
  ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
  ΓΙΑ Ι ΑΠΟ 2 ΜΕΧΡΙ 5
    ΓΙΑ Ξ ΑΠΟ 5 ΜΕΧΡΙ Ι ΜΕ_ΒΗΜΑ -1
      ΑΝ ΟΝ[Ξ] < ΟΝ[Ξ - 1] ΤΟΤΕ
        Τ <- ΟΝ[Ξ] 
        ΟΝ[Ξ] <- ΟΝ[Ξ - 1] 
        ΟΝ[Ξ - 1] <- Τ
      ΤΕΛΟΣ_ΑΝ
    ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
  ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
  ΓΙΑ Ι ΑΠΟ 1 ΜΕΧΡΙ 5
    ΓΙΑ Ξ ΑΠΟ 1 ΜΕΧΡΙ 5
      ΑΝ Ι <> Ξ ΤΟΤΕ
        ΓΡΑΨΕ "ΔΩΣΕ ΑΠΟΤΕΛΕΣΜΑ ", ΟΝ[Ι], " ----- ", ΟΝ[Ξ] 
        ΔΙΑΒΑΣΕ ΑΠΟΤ[Ι, Ξ] 
        ΟΣΟ ΑΠΟΤ[Ι, Ξ] <> "Χ" ΚΑΙ ΑΠΟΤ[Ι, Ξ] <> "1" ΚΑΙ ΑΠΟΤ[Ι, Ξ] <> "2" ΕΠΑΝΑΛΑΒΕ
          ΓΡΑΨΕ "ΛΑΘΟΣ ΑΠΟΤΕΛΕΣΜΑ!!ΔΩΣΕ 1/Χ/2"
          ΔΙΑΒΑΣΕ ΑΠΟΤ[Ι, Ξ] 
        ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
        ΑΓΩΝΕΣ <- ΑΓΩΝΕΣ + 1
        ΑΝ ΑΠΟΤ[Ι, Ξ] = "Χ" ΤΟΤΕ
          ΙΣΟΠΑΛΙΕΣ <- ΙΣΟΠΑΛΙΕΣ + 1
        ΤΕΛΟΣ_ΑΝ
      ΤΕΛΟΣ_ΑΝ
    ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
  ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
  ΓΡΑΨΕ "ΤΟ ΠΟΣΟΣΤΟ ΙΣΟΠΑΛΙΩΝ ΕΙΝΑΙ ", 100*ΙΣΟΠΑΛΙΕΣ/ΑΓΩΝΕΣ, "%"
  ΓΡΑΨΕ "ΔΩΣΕ ΜΟΥ 2 ΟΜΑΔΕΣ ΝΑ ΣΟΥ ΠΩ ΤΑ ΜΕΤΑΞΥ ΤΟΥΣ ΑΠΟΤΕΛΕΣΜΑΤΑ"
  ΔΙΑΒΑΣΕ ΟΝ1, ΟΝ2
  ΓΙΑ Ι ΑΠΟ 1 ΜΕΧΡΙ 5
    ΓΙΑ Ξ ΑΠΟ 1 ΜΕΧΡΙ 5
      ΑΝ Ι <> Ξ ΚΑΙ ΟΝ1 = ΟΝ[Ι] ΚΑΙ ΟΝ2 = ΟΝ[Ξ] ΤΟΤΕ
        ΓΡΑΨΕ ΟΝ[Ι], " ---- ", ΟΝ[Ξ], " : ", ΑΠΟΤ[Ι, Ξ] 
        ΓΡΑΨΕ ΟΝ[Ξ], " ---- ", ΟΝ[Ι], " : ", ΑΠΟΤ[Ξ, Ι] 
        Ξ <- 6
        Ι <- 6
      ΤΕΛΟΣ_ΑΝ
    ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
  ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
  ΓΡΑΨΕ "ΟΜΑΔΕΣ ΜΕ ΕΛΑΧΙΣΤΑ Χ ΩΣ ΓΗΠΕΔΟΥΧΕΣ: "
  ΚΑΛΕΣΕ ΜΙΝΧ_ΓΗΠ(ΑΠΟΤ, ΟΝ, ΙΣΟΠΑΛΙΕΣΕΔ, ΙΣΟΠΑΛΙΕΣΕΚΤΟΣ) 
  ΓΡΑΨΕ "Η ΟΜΑΔΑ ", ΟΝ[5], " ΕΧΕΙ ", ΝΙΚΕΣΟΜΑΔΑΣΕΝΕΚ(1, ΑΠΟΤ, 5), " ΝΙΚΕΣ ΕΝΤΟΣ ΕΔΡΑΣ"
  ΚΑΛΕΣΕ ΒΑΘΜΟΛΟΓΙΑ(ΑΠΟΤ, ΟΝ, ΒΑΘΜ) 
  ΓΙΑ Ι ΑΠΟ 2 ΜΕΧΡΙ 5
    ΓΙΑ Ξ ΑΠΟ 5 ΜΕΧΡΙ Ι ΜΕ_ΒΗΜΑ -1
      ΑΝ ΒΑΘΜ[Ξ] > ΒΑΘΜ[Ξ - 1] ΤΟΤΕ
        Λ <- ΒΑΘΜ[Ξ] 
        ΒΑΘΜ[Ξ] <- ΒΑΘΜ[Ξ - 1] 
        ΒΑΘΜ[Ξ - 1] <- Λ
        Τ <- ΟΝ[Ξ] 
        ΟΝ[Ξ] <- ΟΝ[Ξ - 1] 
        ΟΝ[Ξ - 1] <- Τ
        ΓΙΑ Μ ΑΠΟ 1 ΜΕΧΡΙ 5
          Τ <- ΑΠΟΤ[Μ, Ξ] 
          ΑΠΟΤ[Μ, Ξ] <- ΑΠΟΤ[Μ, Ξ - 1] 
          ΑΠΟΤ[Μ, Ξ - 1] <- Τ
        ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
        ΓΙΑ Μ ΑΠΟ 1 ΜΕΧΡΙ 5
          Τ <- ΑΠΟΤ[Ξ, Μ] 
          ΑΠΟΤ[Ξ, Μ] <- ΑΠΟΤ[Ξ - 1, Μ] 
          ΑΠΟΤ[Ξ - 1, Μ] <- Τ
        ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
      ΤΕΛΟΣ_ΑΝ
    ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
  ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
  ΓΙΑ Ι ΑΠΟ 1 ΜΕΧΡΙ 3
    ΓΡΑΨΕ Ι, "η Η ΟΜΑΔΑ: ", ΟΝ[Ι], " ΜΕ ", ΒΑΘΜ[Ι], " ΒΑΘΜΟΥΣ"
  ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
  ΚΑΛΕΣΕ ΜΙΝΧ_ΓΗΠ(ΑΠΟΤ, ΟΝ, ΙΣΟΠΑΛΙΕΣΕΔ, ΙΣΟΠΑΛΙΕΣΕΚΤΟΣ) 

  ΓΡΑΨΕ "ΑΝ Η ΠΡΩΤΑΘΛΗΤΡΙΑ ΟΜΑΔΑ ", ΟΝ[1], " ΕΠΑΙΡΝΕ ΝΙΚΕΣ ΟΠΟΥ ΠΗΡΕ Χ ΘΑ ΕΙΧΕ"
  ΓΡΑΨΕ ΒΑΘΜ[1] + 2*(ΙΣΟΠΑΛΙΕΣΕΔ[1] + ΙΣΟΠΑΛΙΕΣΕΚΤΟΣ[1]) 
  ΓΡΑΨΕ "ΑΚΟΜΑ ΕΧΕΙ ΝΙΚΗΣΕΙ ΣΕ ", ΝΙΚΕΣΟΜΑΔΑΣΕΝΕΚ(0, ΑΠΟΤ, 1), " ΑΓΩΝΕΣ ΕΚΤΟΣ ΕΔΡΑΣ"
  ΓΡΑΨΕ "___________________________________________"
  ΓΡΑΨΕ "Η ΔΕΥΤΕΡΗ ΟΜΑΔΑ ", ΟΝ[2] 
  ΓΡΑΨΕ "ΠΗΡΕ ", ΙΣΟΠΑΛΙΕΣΕΚΤΟΣ[2] + ΙΣΟΠΑΛΙΕΣΕΔ[2], " ΙΣΟΠΑΛΙΕΣ ΣΤΟ ΠΡΩΤΑΘΛΗΜΑ"
  ΓΡΑΨΕ "ΑΚΟΜΑ ΕΧΕΙ ΑΗΤΤΗΤΟ ΕΚΤΟΣ ΕΔΡΑΣ ΣΕ ", ΙΣΟΠΑΛΙΕΣΕΚΤΟΣ[2] + ΝΙΚΕΣΟΜΑΔΑΣΕΝΕΚ(0, ΑΠΟΤ, 2), " ΑΓΩΝΕΣ"
  ΓΡΑΨΕ "__________________________"
  ΓΡΑΨΕ ΟΝ[1], " ----- ", ΟΝ[3], " ΔΙΑΦΟΡΑ ΒΑΘΜΩΝ ΠΡΩΤΟΥ ΜΕ ΤΡΙΤΟ: ", ΒΑΘΜ[1] - ΒΑΘΜ[3] 
  ΓΡΑΨΕ ΟΝ[1], " ----- ", ΟΝ[3], " : ", ΑΠΟΤ[1, 3] 
  ΓΡΑΨΕ ΟΝ[3], " ----- ", ΟΝ[1], " : ", ΑΠΟΤ[3, 1] 
  ΓΡΑΨΕ "_____________________________"
  ΓΡΑΨΕ ΟΝ[2], " ----- ", ΟΝ[3], " ΔΙΑΦΟΡΑ ΒΑΘΜΩΝ ΔΕΥΤΕΡΟΥ ΜΕ ΤΡΙΤΟ: ", ΒΑΘΜ[2] - ΒΑΘΜ[3] 
  ΓΡΑΨΕ "______________________________________________________"
  ΑΡΧΗ_ΕΠΑΝΑΛΗΨΗΣ
    ΓΡΑΨΕ "ΔΩΣΕ ΜΟΥ ΟΝΟΜΑ ΟΜΑΔΑΣ ΝΑ ΣΟΥ ΠΩ ΚΑΤΑΤΑΞΗ"
    ΔΙΑΒΑΣΕ ΟΝ1
    Β <- -1
    Λ <- 1
    ΓΙΑ Ι ΑΠΟ 1 ΜΕΧΡΙ 5
      ΑΝ ΟΝ1 = ΟΝ[Ι] ΤΟΤΕ
        Β <- ΒΑΘΜ[Ι] 
        ΓΡΑΨΕ "Η ΟΜΑΔΑ ", ΟΝ[Ι], " ΕΙΝΑΙ ", Ι, "η"
        Θ <- Ι
      ΤΕΛΟΣ_ΑΝ
    ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
    ΑΝ Β <> -1 ΤΟΤΕ
      ΓΙΑ Ι ΑΠΟ 1 ΜΕΧΡΙ 5
        ΑΝ ΒΑΘΜ[Ι] = Β ΚΑΙ ΟΝ[Ι] <> ΟΝ1 ΤΟΤΕ
          ΑΝ Λ = 1 ΤΟΤΕ
            ΓΡΑΨΕ "ΜΑΖΙ ΜΕ ΤΙΣ/ΤΗΝ:", ΟΝ[Ι] 
            Λ <- 0
          ΑΛΛΙΩΣ
            ΓΡΑΨΕ ΟΝ[Ι] 
          ΤΕΛΟΣ_ΑΝ
        ΤΕΛΟΣ_ΑΝ
      ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
      ΓΡΑΨΕ "ΤΟ ΠΟΣΟΣΤΟ ΝΙΚΩΝ ΕΝΤΟΣ ΕΔΡΑΣ ΤΗΣ ", ΟΝ1, " ΕΙΝΑΙ ", 100*ΝΙΚΕΣΟΜΑΔΑΣΕΝΕΚ(1, ΑΠΟΤ, Θ)/4, "%"
    ΑΛΛΙΩΣ
      ΓΡΑΨΕ "ΔΕΝ ΥΠΑΡΧΕΙ ΤΕΤΟΙΑ ΟΜΑΔΑ ΣΤΗΝ ΚΑΤΑΤΑΞΗ"
    ΤΕΛΟΣ_ΑΝ
  ΜΕΧΡΙΣ_ΟΤΟΥ(Β <> -1) 
ΤΕΛΟΣ_ΠΡΟΓΡΑΜΜΑΤΟΣ




ΔΙΑΔΙΚΑΣΙΑ ΜΙΝΧ_ΓΗΠ(ΑΠΟΤ, ΟΝ, ΙΣΟΠΑΛΙΕΣΕΔ, ΙΣΟΠΑΛΙΕΣΕΚΤΟΣ) 

ΜΕΤΑΒΛΗΤΕΣ
  ΑΚΕΡΑΙΕΣ: ΙΣΟΠΑΛΙΕΣΕΔ[5], ΙΣΟΠΑΛΙΕΣΕΚΤΟΣ[5], Ι, Ξ, ΜΙΝΧΓΗΠ
  ΧΑΡΑΚΤΗΡΕΣ: ΜΙΝΧΟΜ, ΟΝ[5], ΑΠΟΤ[5, 5] 
ΑΡΧΗ
  ΜΙΝΧΓΗΠ <- 5*5
  ΓΙΑ Ι ΑΠΟ 1 ΜΕΧΡΙ 5
    ΙΣΟΠΑΛΙΕΣΕΔ[Ι] <- 0
    ΙΣΟΠΑΛΙΕΣΕΚΤΟΣ[Ι] <- 0
    ΓΙΑ Ξ ΑΠΟ 1 ΜΕΧΡΙ 5
      ΑΝ ΑΠΟΤ[Ι, Ξ] = "Χ" ΤΟΤΕ
        ΙΣΟΠΑΛΙΕΣΕΔ[Ι] <- ΙΣΟΠΑΛΙΕΣΕΔ[Ι] + 1
      ΤΕΛΟΣ_ΑΝ
      ΑΝ ΑΠΟΤ[Ξ, Ι] = "Χ" ΤΟΤΕ
        ΙΣΟΠΑΛΙΕΣΕΚΤΟΣ[Ι] <- ΙΣΟΠΑΛΙΕΣΕΚΤΟΣ[Ι] + 1
      ΤΕΛΟΣ_ΑΝ
    ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
    ΑΝ ΙΣΟΠΑΛΙΕΣΕΔ[Ι] < ΜΙΝΧΓΗΠ ΤΟΤΕ
      ΜΙΝΧΟΜ <- ΟΝ[Ι] 
      ΜΙΝΧΓΗΠ <- ΙΣΟΠΑΛΙΕΣΕΔ[Ι] 
    ΤΕΛΟΣ_ΑΝ
  ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ

  ΓΙΑ Ι ΑΠΟ 1 ΜΕΧΡΙ 5
    ΑΝ ΜΙΝΧΓΗΠ = ΙΣΟΠΑΛΙΕΣΕΔ[Ι] ΤΟΤΕ
      ΓΡΑΨΕ ΟΝ[Ι] 
    ΤΕΛΟΣ_ΑΝ
  ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
ΤΕΛΟΣ_ΔΙΑΔΙΚΑΣΙΑΣ

ΣΥΝΑΡΤΗΣΗ ΝΙΚΕΣΟΜΑΔΑΣΕΝΕΚ(Λ, ΑΠΟΤ, Ι): ΑΚΕΡΑΙΑ
ΜΕΤΑΒΛΗΤΕΣ
  ΑΚΕΡΑΙΕΣ: Ν1, Ν2, Λ, Ι, Ξ
  ΧΑΡΑΚΤΗΡΕΣ: ΑΠΟΤ[5, 5] 
ΑΡΧΗ
  Ν1 <- 0
  Ν2 <- Ν1

  ΓΙΑ Ξ ΑΠΟ 1 ΜΕΧΡΙ 5
    ΑΝ ΑΠΟΤ[Ι, Ξ] = "1" ΤΟΤΕ
      Ν1 <- Ν1 + 1
    ΤΕΛΟΣ_ΑΝ
    ΑΝ ΑΠΟΤ[Ξ, Ι] = "2" ΤΟΤΕ
      Ν2 <- Ν2 + 1
    ΤΕΛΟΣ_ΑΝ
  ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
  ΑΝ Λ = 1 ΤΟΤΕ
    ΝΙΚΕΣΟΜΑΔΑΣΕΝΕΚ <- Ν1
  ΑΛΛΙΩΣ
    ΝΙΚΕΣΟΜΑΔΑΣΕΝΕΚ <- Ν2
  ΤΕΛΟΣ_ΑΝ
ΤΕΛΟΣ_ΣΥΝΑΡΤΗΣΗΣ

ΔΙΑΔΙΚΑΣΙΑ ΒΑΘΜΟΛΟΓΙΑ(ΑΠΟΤ, ΟΝ, ΒΑΘΜ) 
ΜΕΤΑΒΛΗΤΕΣ
  ΑΚΕΡΑΙΕΣ: Ι, Ξ, ΒΑΘΜ[5] 
  ΧΑΡΑΚΤΗΡΕΣ: ΑΠΟΤ[5, 5], ΟΝ[5] 
ΑΡΧΗ

  ΓΙΑ Ι ΑΠΟ 1 ΜΕΧΡΙ 5
    ΒΑΘΜ[Ι] <- 0
    ΓΙΑ Ξ ΑΠΟ 1 ΜΕΧΡΙ 5
      ΑΝ ΑΠΟΤ[Ι, Ξ] = "1" ΤΟΤΕ
        ΒΑΘΜ[Ι] <- ΒΑΘΜ[Ι] + 3
      ΤΕΛΟΣ_ΑΝ
      ΑΝ ΑΠΟΤ[Ξ, Ι] = "2" ΤΟΤΕ
        ΒΑΘΜ[Ι] <- ΒΑΘΜ[Ι] + 3
      ΤΕΛΟΣ_ΑΝ
      ΑΝ ΑΠΟΤ[Ι, Ξ] = "Χ" ΤΟΤΕ
        ΒΑΘΜ[Ι] <- ΒΑΘΜ[Ι] + 1
      ΤΕΛΟΣ_ΑΝ
      ΑΝ ΑΠΟΤ[Ξ, Ι] = "Χ" ΤΟΤΕ
        ΒΑΘΜ[Ι] <- ΒΑΘΜ[Ι] + 1
      ΤΕΛΟΣ_ΑΝ
    ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
  ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ


ΤΕΛΟΣ_ΔΙΑΔΙΚΑΣΙΑΣ



